Linear Recurrence Relations in Maple

Date: Nov 7, 2001
Last Revision: Nov 15, 2001
Maple 6

Bent E. Petersen
bent@alum.mit.edu
petersen@math.orst.edu

Course: Mth 355 (a.k.a. Mth 399)
Term: Fall 2001
File name: `355f2001_recur.mws`
Assignment 6 (problems at end) - due Nov 14, 2001.

This worksheet contains a few comments on some of Maple's recurrence relations support. It composed fairly quickly and is unlikely to be free of errors. If you find a serious error be sure to mention it in your solution report.

```
> restart;
```

Let's load the plots package - just in case.

```
> with(plots):
Warning, the name changecoords has been redefined
```

Example 1

The Maple function rsolve() is used to solve linear recurrence relations. Here is an example of an order 2 recurrence relation with initial conditions.

```
> eqn1:=a(n)=a(n-1)+a(n-2);   # Fibonacci
```

$$eqn1 := \mathrm{a}(n) = \mathrm{a}(n-1) + \mathrm{a}(n-2)$$

```
> init1:=a(0)=0,a(1)=1;
```

$$init1 := \mathrm{a}(0) = 0,\ \mathrm{a}(1) = 1$$

To solve we throw all the equations into one set and also specify for which variable to solve.

```
> soln1:=rsolve({eqn1,init1},a);
```

$$soln1 := \frac{\left(\frac{1}{5}\sqrt{5}-1\right)\left(-2\frac{1}{1-\sqrt{5}}\right)^n}{1-\sqrt{5}} + \frac{\left(-\frac{1}{5}\sqrt{5}-1\right)\left(-2\frac{1}{1+\sqrt{5}}\right)^n}{1+\sqrt{5}}$$

That's all there is to it! This expression is not in the usual Binet form, so you may want to try to get Maple to cast it in a more familiar form. Good luck!

We can also use rsolve() to find the generating function for the solution. Note the single quotes surrounding genfunc in the following command.

```
> gen1:=rsolve({eqn1,init1},a,'genfunc'(z));
```

$$gen1 := -\frac{z}{-1+z+z^2}$$

We can even define a procedure which will compute the a(n).

```
> fun1:=rsolve({eqn1,init1},a,'makeproc'):
```

The fun1() procedure can be use to plot the sequence a(n)

```
> sq1:=seq([n,fun1(n)],n=0..16);
```

$sq1$:= [0, 0], [1, 1], [2, 1], [3, 2], [4, 3], [5, 5], [6, 8], [7, 13], [8, 21], [9, 34], [10, 55], [11, 89], [12, 144], [13, 233], [14, 377], [15, 610], [16, 987]

```
> PLOT(POINTS(sq1),SYMBOL(BOX,16),TEXT([5,1000],"fun1 - Fibonacci
  sequence"));
```



We could plot a curve instead

```
> PLOT(CURVES([sq1]),TEXT([5,900],"fun1 - Fibonacci sequence"));
```



Here's one (kludgy) way to get the characteristic polynomial and the characteristic root

```
> subs(a(n)=x^n,a(n-1)=x^(n-1),a(n-2)=x^(n-2),eqn1)/x^(n-2):
  cpoly1:=simplify(%);
```

$$cpoly1 := x^2 = x + 1$$

```
> croot1:=solve(cpoly1,x);
```

$$croot1 := \frac{1}{2}+\frac{1}{2}\sqrt{5}, \frac{1}{2}-\frac{1}{2}\sqrt{5}$$

Note it is not too difficult to write a procedure to compute the order of the recurrence equation and then to compute the characteristic polynomial. I will do this in the next worksheet (Assignment 7).

Example 2

Here's an inhomogeneous equation related to the Fibonacci equation

```
> eqn2:=a(n)=a(n-1)+a(n-2)-n;
```

$$eqn2 := a(n) = a(n-1) + a(n-2) - n$$

```
> init2:=a(0)=0,a(1)=1;
```

$$init2 := a(0) = 0, a(1) = 1$$

```
> soln2:=rsolve({eqn2,init2},a);
```

$$soln2 := \frac{\left(\frac{1}{5}\sqrt{5}-1\right)\left(-2\frac{1}{1-\sqrt{5}}\right)^n}{1-\sqrt{5}} + \frac{\left(-\frac{1}{5}\sqrt{5}-1\right)\left(-2\frac{1}{1+\sqrt{5}}\right)^n}{1+\sqrt{5}} + n + 3$$
$$+ \frac{(1+\sqrt{5})\left(-2\frac{1}{1-\sqrt{5}}\right)^n}{1-\sqrt{5}} + \frac{(1-\sqrt{5})\left(-2\frac{1}{1+\sqrt{5}}\right)^n}{1+\sqrt{5}}$$

```
> gen2:=rsolve({eqn2,init2},a,'genfunc'(z));
```

$$gen2 := -\frac{\left(-2\frac{1}{1-z}-\frac{z}{(1-z)^2}\right)z^2+z}{-1+z+z^2}$$

```
> fun2:=rsolve({eqn2,init2},a,'makeproc'):
> sq2:=seq([n,fun2(n)],n=0..20):
> PLOT(POINTS(evalf(sq2)),SYMBOL(DIAMOND,16),TEXT([5,-20000],"fun2")
  );
```



Example 3

```
> eqn3:=a(n)=a(n-1)-a(n-2)/2;
```

$$eqn3 := a(n) = a(n-1) - \frac{1}{2} a(n-2)$$

```
> init3:=a(0)=1,a(1)=1;
```

$$init3 := a(0) = 1, a(1) = 1$$

```
> soln3:=rsolve({eqn3,init3},a);
```

$$soln3 := \left(\frac{1}{2} + \frac{1}{2} I\right)\left(\frac{1}{2} - \frac{1}{2} I\right)^n + \left(\frac{1}{2} - \frac{1}{2} I\right)\left(\frac{1}{2} + \frac{1}{2} I\right)^n$$

```
> gen3:=rsolve({eqn3,init3},a,'genfunc'(z));
```

$$gen3 := 2 \frac{1}{2 - 2z + z^2}$$

```
> fun3:=rsolve({eqn3,init3},a,'makeproc'):
> sq3:=seq([n,fun3(n)],n=0..20):
> PLOT(POINTS(evalf(sq3)),SYMBOL(DIAMOND,16),TEXT([20,-1],"fun3"));
```




```
> PLOT(CURVES([evalf(sq3)]),SYMBOL(DIAMOND,16),TEXT([4,0.9],"fun3"))
;
```




```
> subs(a(n)=x^n,a(n-1)=x^(n-1),a(n-2)=x^(n-2),eqn3)/x^(n-2):
cpoly3:=simplify(%);
```

$$cpoly3 := x^2 = x - \frac{1}{2}$$

```
> croot3:=solve(cpoly3,x);
```

$$croot3 := \frac{1}{2} + \frac{1}{2} I, \frac{1}{2} - \frac{1}{2} I$$

Example 4

```
> eqn4:=a(n)=-a(n-2);
```

$$eqn4 := \mathrm{a}(n) = -\mathrm{a}(n-2)$$

```
> init4:=a(0)=2,a(1)=-1;
```

$$init4 := \mathrm{a}(0) = 2, \mathrm{a}(1) = -1$$

```
> soln4:=rsolve({eqn4,init4},a);
```

$$soln4 := \left(1 - \frac{1}{2} I\right)(-I)^n + \left(1 + \frac{1}{2} I\right) I^n$$

```
> gen4:=rsolve({eqn4,init4},a,'genfunc'(z));
```

$$gen4 := \frac{2-z}{1+z^2}$$

```
> fun4:=rsolve({eqn4,init4},a,'makeproc'):
> sq4:=seq([n,fun4(n)],n=0..20):
> PLOT(POINTS(evalf(sq4)),SYMBOL(DIAMOND,20),TEXT([3,3],"fun4"));
```



```
> PLOT(CURVES([evalf(sq4)]),SYMBOL(DIAMOND,16),TEXT([2,1.8],"fun3"))
  ;
```



Here we have a periodic solution.

```
> subs(a(n)=x^n,a(n-1)=x^(n-1),a(n-2)=x^(n-2),eqn4)/x^(n-2):
  cpoly4:=simplify(%);
```

$$cpoly4 := x^2 = -1$$

```
> croot4:=solve(cpoly4,x);
```

$$croot4 := I, -I$$

Example 5

```
> eqn5:=a(n)=3*a(n-1)+4*a(n-2)-12*a(n-3);
```

$$eqn5 := a(n) = 3\ a(n-1) + 4\ a(n-2) - 12\ a(n-3)$$

```
> subs(seq(a(n-k)=x^(n-k),k=0..3),eqn5)/x^(n-3):
  cpoly5:=simplify(%);
```

$$cpoly5 := x^3 = 3\ x^2 + 4\ x - 12$$

```
> croot5:=solve(cpoly5,x);
```

$$croot5 := -2, 2, 3$$

```
> init5:=a(0)=2,a(1)=5,a(2)=13;
```

$$init5 := a(0) = 2, a(1) = 5, a(2) = 13$$

```
> soln5:=rsolve({eqn5,init5},a);
```

$$soln5 := 3^n + 2^n$$

Example 6

Let a(n) be the number of strings of 0's and 1's of length n which do not contain the bit pattern 11. If a such a string starts with 0 the remainder can be any string of length n-1 which does not contain 11. If on the other hand it starts with 1 then the second character must be 0 and after that we can have any string of length n-2 which does not contain 11. Thus (note we just have a shifted Fibonacci sequence)

```
> eqn6:=a(n)=a(n-1)+a(n-2);
```

$$eqn6 := a(n) = a(n-1) + a(n-2)$$

```
> init6:=a(0)=1,a(1)=2;
```

$$init6 := a(0) = 1, a(1) = 2$$

```
> soln6:=rsolve({eqn6,init6},a(n));
```

$$soln6 := \frac{\left(-\frac{1}{5}\sqrt{5}-1\right)\left(-2\frac{1}{1-\sqrt{5}}\right)^n}{1-\sqrt{5}} + \frac{\left(\frac{1}{5}\sqrt{5}-1\right)\left(-2\frac{1}{1+\sqrt{5}}\right)^n}{1+\sqrt{5}}$$

```
> A6:=unapply(round(soln6),n);
```

$$A6 := n \rightarrow \mathrm{round}\left(\frac{\left(-\frac{1}{5}\sqrt{5}-1\right)\left(-2\frac{1}{1-\sqrt{5}}\right)^n}{1-\sqrt{5}} + \frac{\left(\frac{1}{5}\sqrt{5}-1\right)\left(-2\frac{1}{1+\sqrt{5}}\right)^n}{1+\sqrt{5}}\right)$$

```
> for k from 0 to 14 do printf("%5d", A6(k)); od;
    1    2    3    5    8   13   21   34   55   89  144  233  377  610  987
```

Example 7

Consider n nonparallel lines in the plane and suppose no three of them meet in a point. The lines divide the plane into a(n) regions. If we already have n-1 lines and add one more line, it cuts each of the existing n-1 lines and so creates n new regions. (This statement may require some argument.) Thus

```
> eqn7:=a(n)=a(n-1)+n;
```

$$eqn7 := a(n) = a(n-1) + n$$

```
> init7:=a(0)=1;
```

$$init7 := a(0) = 1$$

```
> soln7:=rsolve({eqn7,init7},a(n));
```

$$soln7 := (n+1)\left(\frac{1}{2}n+1\right)-n$$

```
> for k from 0 to 20 do printf("%4d", subs(n=k,soln7)); od;
   1   2   4   7  11  16  22  29  37  46  56  67  79  92 106 121 137 154 172 191 2
11
> rsolve({eqn7,init7},a,'genfunc'(z)): gen7:=simplify(%);
```

$$gen7 := -\frac{-z+1+z^2}{(-1+z)^3}$$

Example 8

Suppose A is a set of cardinality n, for convenience say, A = {1,2,...,n}. Let S(n,m) be the number of partitions of A into m subsets (so the Stirling number of the second kind).

The partitions which do contain the set {n} correspond to partitions of {1,2,...,n-1} into m-1 subsets. There are S(n-1,m-1) such partitions.

The partitions which do not contain {n} correspond to partitions of {1,2,...,n-1} into m subsets any one of which we can add n to in order to obtain a partition of {1,2,...,n}. There are m S(n-1,m) such partitions. Thus we have

```
> eqn8:=S(n,m)=S(n-1,m-1)+m*S(n-1,m);
```

$$eqn8 := S(n, m) = S(n-1, m-1) + m\, S(n-1, m)$$

```
> init8:=S(n,1)=1,S(n,n)=1;
```

$$init8 := S(n, 1) = 1,\ S(n, n) = 1$$

Here we have an example of a double recurrence relation. Maple does not seem to be able to handle such recurrences..

Example 9

Consider a string of length n+1. The number of ways we can insert n pairs of parantheses so as to form n "products" of pairs of characters in the string is Cat(n) (a Catalan number). Cat(n) satisfies a first order recurrence equation.

```
> eqn9:=(n+1)*c(n)=(4*n-2)*c(n-1);
```

$$eqn9 := (n+1)\, c(n) = (4\, n-2)\, c(n-1)$$

```
> init9:=c(1)=1;
```

$$init9 := c(1) = 1$$

```
> rsolve({eqn9,init9},c(n)): Cat:=unapply(%,n);
```

$$Cat := n \to \frac{4^n \,\Gamma\left(n+\frac{1}{2}\right)}{\Gamma(n+2)\sqrt{\pi}}$$

It is difficult to directly simplify this expression. However we can get Maple to verify that

```
> 'Cat(n)' = 1/(n+1)*binomial(2*n,n);
```

$$\mathrm{Cat}(n) = \frac{\mathrm{binomial}(2\,n, n)}{n+1}$$

Here's the verification - actually an automated proof of sorts!

```
> simplify(Cat(n)-binomial(2*n,n)/(n+1));
```

$$0$$

Problems

Solve the following recursive equations. In each case find the generating function as well.

Problem 1

```
> a(n) = 2*a(n-1)-7; a(0)=5;
```

$$a(n) = 2\,a(n-1) - 7$$

$$a(0) = 5$$

Problem 2

```
a(n) = 7*a(n-1)-11*a(n-2); a(0)=3,a(1)=-2;
```

$$a(n) = 7\,a(n-1) - 11\,a(n-2)$$

$$a(0) = 3,\ a(1) = -2$$

Problem 3

```
> a(n) = a(n-1) + n^3; a(1)=1;
```

$$a(n) = a(n-1) + n^3$$

$$a(1) = 1$$

What does this problem have to do with

```
> Sum(k^3,k=1..n);
```

$$\sum_{21=1}^{n} 9261$$

What does it have to do with

```
> Sum(k,k=1..n)^2;
```

$$\left(\sum_{21=1}^{n} 21\right)^2$$

Problem 4

```
> a(n) = a(n-1) + n^5; a(1)=1;
```

$$a(n) = a(n-1) + n^5$$
$$a(1) = 1$$

What does this problem have to do with

```
> Sum(k^5,k=1..n);
```

$$\sum_{21=1}^{n} 4084101$$

Problem 5

```
> a(n) = 2*a(n-1)+11*a(n-2)-12*a(n-3)+5*2^n; a(0)=-2,a(1)=3,a(2)=1;
```

$$a(n) = 2\,a(n-1) + 11\,a(n-2) - 12\,a(n-3) + 5\,2^n$$
$$a(0) = -2,\ a(1) = 3,\ a(2) = 1$$

Problem 6

```
> b(n)=b(n-1)/b(n-2); b(0)=2,b(1)=4;
```

$$b(n) = \frac{b(n-1)}{b(n-2)}$$
$$b(0) = 2,\ b(1) = 4$$

Hint: Try a(n) = log(b(n)).

Compute b(10) from your solution and compare with b[10] computed by

```
> b[0]:=2: b[1]:=4: for k from 2 to 10 do b[k]:=b[k-1]/b[k-2]; od:
```

Problem 7

Maple will solve some recursive equations which are of infinite order. For example, try

```
> a(2*n) = a(n)+7; a(1)=0;
```

$$a(2\,n) = a(n) + 7$$
$$a(1) = 0$$

```
>
```